
==============================Simple_PLL=================================

CREATE A FULL BEHAVIORAL PLL USING LESS THAN TEN LINES OF SPICE. 

Simple_PLL
*                                   ______________________________
*                                   |                             \
*                                   | V = sin(2*PI*Fre*T+IntFM*T)  \
*           VMULT    RLP      LPASS |                        ^      \
*              ____/\  /\  /\_______|     RINT     VPHAS    /|\      \
*            _|_     \/  \/   |     | __/\  /\  /\_____ _ _ _|        \_ LOSC
*   VIN     /   \            _|_    |     \/  \/  _|_                 / |
*    ____|\/  \/ \     CLP   ___    | FM          ___      SYSTEM    /  |
*  _|_   |/\  /\ /           _|_    |       CINT  _|_      MODELED  /   |
* /_  \     \___/ BMULT      ///    |             ///        VCO   /    |
*// \  \      ^                     |_____________________________/     |
*\   \//     /_\                                                        |
* \___/       |_________________________________________________________|
*  _|_
*  ///   WinSpice/MacSpice tested 2.19.11_6.58PM dsauersanjose@aol.com  
*
*=========All_System_Modeled_PLL==================

VT        VT     0      DC     0      PWL(   0      0      10      10      )
BVIN      VIN    0      V =           sin( 6.28*1000*V(VT) ) 
BLOSC     LOSC   0      V =           sin( 6.28*1000*V(VT)  + 10*V(VPHAS)*V(VT) )
BMULT     VMULT  0      V =           V(VIN)*V(LOSC)
RLP       VMULT  LPASS  1k
CLP       LPASS  0      1.6u
RINT      LPASS  VPHAS  1meg
CINT      VPHAS  0      1u

.control
set       pensize =     2
tran      .1m    5      0     .1m
plot      lpass 
plot      vin    lpass  losc  vmult  xlimit  9m    10m
plot      vin    lpass  losc  vmult  xlimit  498m  500m
plot      vin    lpass  losc  vmult  xlimit  4998m 5000m
.endc
.end


